
Introduction 

 
Whilst Excel is famous for its numeric calculations and presentation abilities it can also do 
some really cool things with text. 
 
This webinar will take you through all the text functions you need to know and how to use 
them effectively. 
 
Excel has a built-in feature that can speed up splitting up text. 
 
We will examine three common things done with text: 
 

1. Split it 
 

2. Join it 
 

3. Tweak it 
 
We will finish with a few ways to use the functions together to solve some common issues. 
 
Quotation Marks 
 
When using text in a formula you need to enclose the text entries in quotation marks. 
 
 
Blanks 
 
To refer to a blank cell in Excel you need to use two consecutive quotation marks  
 

"" 

 
 
Spaces 
 
When working with text, always remember that a space is a text character. You need to take 
spaces into consideration when working with text in terms of splitting and joining text. To 
refer to a space you use a space between two quotation marks.  
 

" " 

 
Text Values 
 
Text functions returns text entries. Sometimes you will need to convert a text number into a 
real number. You will learn an easy technique to achieve that. 

Splitting Text 
 
When data comes out of other systems it may not be in a form you want or can use for the 

output you require.  The text features and functions allow you to modify the data so that is 

usable. 

 
A common requirement is to split the first and last name.  

Text To Columns 
 
If your data has some structure or uses specific characters to 
separate the text strings then Text to Columns may provide 
the quick solution you need to split up your data. We will use 
a small range but it can work on thousands of rows.  
 
WARNING: Make sure there are no entries on the right of the 
data you want to split. 
 
PRACTICE: 
 

1. In this example we have a first and last name in column A. 
 

2. To use Text to Columns you need to first select the range to work with. In our case 
select A2:A6. 
 

3. Click the Data ribbon tab and click the Text to 
Columns icon. 
 

4. The first step in a three step Wizard opens. 
 

5. In most cases you will use the Delimited option which 
assumes a character is used to split up the data.  
 
In our case it is a space character. If your data has 
fixed widths you can use the Fixed width options. 
 

6. Click the Next button. 

7. In Step 2 change the option to 

Space and you will see the data 
change in the bottom of the 
dialog to show the split. 
 
You can click the Finish button at 
this stage if you have no numbers 
involved.  
 
If you have numbers like bank 
account or mobile phone 
numbers you will want to use the 
third step in the Wizard. 
 
Click the Next button. 
 

8. In the third step of the Wizard you 
can select a column and apply a 
format to them.  
 
Applying a Text format to a 
column of bank account or mobile 
numbers will retain any leading 
zeroes. If you don’t use the Text 
option the leading zeroes will be 
lost.   
 
Click Finish button to see the 
result.    

Text Functions 
 

 

LEN Function 

 

One of the simplest of the text functions is the LEN function. It is short for length and returns 

how many characters are in the cell’s value. 

=LEN(Cell_ Reference) 

Example 

=LEN(A2) 

Will return how many characters (including spaces) are in the value in A2. 

TIP: 

Don’t forget spaces are characters too. A space counts as a character. They need to be 

considered when handling text. Some systems download data into Excel that contains 

leading or trailing spaces. See TRIM function below 

 

TRIM Function 

 

Another simple to use function that performs a useful service. It removes all leading and 

trailing space characters whilst leaving spaces within the text alone. Leading and trailing 

spaces in data are a common problem for VLOOKUP, SUMIF and SUMIFS functions.   

=TRIM(Cell_ Reference) 

Example 

=TRIM(A2) 

 

Extracting Text via Formulas 
 
 
There are three functions that can extract text from cells. We can extract from the left, the 
right and the middle of the cell.  

LEFT and RIGHT Functions 

 

These two functions work the same way, but from either end of the text string. They extract 

a specified number of characters from a text string from either the left or the right of the 

text string.  

=LEFT(Text, Number_of_Characters) 

=RIGHT(Text, Number_of_Characters) 

Text is typically a cell reference.  

Number_of_Characters (optional) can by typed in (hard coded) or a cell reference or a 

function or formula. If omitted, 1 is used. 

Examples 

If A1 contains the text John Smith 

Then  

=LEFT(A1,4) will display John 

=RIGHT(A1,4) will display mith 

If you omit the Number_of_Characters then it will default to the first (LEFT) or last (RIGHT) 

characters. 

 

MID Function 

 

The MID function allows you to extract text from the middle of a text string.  

=MID(Text, Start_Character_Number, Number of Characters)  

Text is typically a cell reference.  

Start_Character_Number  is the character position where the extraction will start from, this 

value can be hard-keyed; a cell link or a formula. This character position is the first character 

of the string that is returned.  

Number of Characters is how many characters are to be extracted. This can be hard-keyed; a 

cell link or a formula. 

Examples 

If A1 contains the text John Smith 

Then 

=MID(A1,6,2) will display Sm  

=MID(A1,2,5) will display ohn S 

 

PRACTICE: 

1. In the MID sheet we can see how MID works to extract text. 

 

2. Cell B2 has the following formula keyed in and copied down 

 

=MID(A2,4,3) 

 

3. Note cell B5 even though it is displaying a number it is left aligned, meaning it is text. 

To convert this to a number you can modify as follows. 

 

=MID(A5,4,3)*1 

 

This will display an error if the text cannot be converted into a number. 

 

4. The formula in cell F2 is  

 

=MID(A2,D2,E2) 

 

You can change the values in columns D and E to modify the text extracted in 

column F. Using a cell reference to define how many characters to extract is useful if 

the data being extracted is a fixed length eg bank account BSB or a product code and 

should these ever be lengthened or shortened referencing a cell reference is a very 

quick method of overcoming the need to re-write the formula in various 

spreadsheets. 

 

5. Note the result in cell F5. You may think an error message would be generated. The 

LEFT, RIGHT and MID functions when used by themselves rarely display errors.     

 

The MID, LEFT and RIGHT functions are often used with the SEARCH function to provide a 

flexible extraction technique. 

SEARCH Function 

 

The SEARCH function looks for one text string within another text string and returns the 

character position of the first letter of the text being searched for. If the text being searched 

for is not in the text string the function returns a #VALUE! error.  

The SEARCH function is not case sensitive. If you need to perform a case sensitive search use 

the FIND function, which has an identical syntax to the SEARCH function 

=SEARCH(Find_Text,Within_Text,Start_Number) 

Find_Text is the text string to search for. It can be keyed-in; a cell reference or a formula. It 

can be a single character or multiple characters. You must enclose all text within quotation 

marks. 

Within_Text is the string to search in  

Start_Number (optional) and determines the character position to start searching from 

Examples 

=SEARCH("a","cat") will display 2 

=SEARCH("cat","top cat") will display 5 

 

PRACTICE: 

1. Open the SEARCH sheet. There are some names that we want to separate into First 

Name and Last Name. 

 

2. The solutions will involves the LEFT, RIGHT, LEN and SEARCH functions 

 

3. In cell B2 enter the following formula  

 

=LEFT(A2,SEARCH(" ",A2)-1) 

 

To extract the first name, which is at the start of this text string, we use the SEARCH 

function to identify where the space is. We subtract 1 from the SEARCH result 

because the number returned will be one more character than we need a it includes 

the space. Deducting one leaves us with the name only, no space.  

 

The formula can be copied down, note the error in the last cell. There is no space in 

the name. 

4. In cell C2 enter the following formula 

 

=RIGHT(A2,LEN(A2)-SEARCH(" ",A2)) 

 

To extract the last name from the end of the string we need the number of 

characters of the last name. This is calculated by subtracting the SEARCH result for 

the space from the length of the string. This calculates how many characters are 

after the space. 

 

Again this will display an error is there is no space in the name. 

 

Handling Errors 
 

As we have seen the SEARCH function can display errors. As a general rule you should avoid 

displaying errors. Your formulas need to handle errors in a manner that doesn’t cause other 

errors 

Excel 2007 added a function that simplifies handling errors. The IFERROR function does not 

work in previous versions. 

 

IFERROR Function 

 

The IFERROR function has a very simple layout.  

=IFERROR(Formula, Value_If_Error) 

Formula – any formula that could return an error 

Value_If_Error – the value, formula or text to display if an error is encountered.     

Example 

=IFERROR(A1/B2,0)   

will display a zero if B2 is zero and avoids the #DIV/0! Error. 

PRACTICE: 

1. Let’s fix the first name and last name formula in SEARCH sheet to handle a single 

word name. 

 

2. Amend the formula in B2 from  

 

=LEFT(A2,SEARCH(" ",A2)-1) 

 

to 

 

=IFERROR(LEFT(A2,SEARCH(" ",A2)-1),A2) 

 

The highlighted section has been added.  

This will treat a name without a space as a first name 

 

3. Amend the formula in C2 to 

 

=IFERROR(RIGHT(A2,LEN(A2)-SEARCH(" ",A2)),"") 

 

The displays a blank cell if there is no space in the name 

 

4. Copy the formulas down to test. 

 
 

Joining Text 
 
Creating headings and even whole sentences is possible with one or two additional functions 
and a certain symbol. 
 
The & symbol (its above the number 7 on the keyboard) will join text together. There is a 
function called CONCATENATE that does it too but its easier to use the &. 
 
This can be used to automate journal narrations, report comments and headings. 

Automating Headings and Text 
 

Report headings often contain dates and some reports have narrations describing results. It 

is possible to automate some of these text entries. We have seen the text functions that 

Excel has. It has one other function that is useful for headings and narrations. The TEXT 

function allows you to format dates and dollar amounts.  

TEXT Function 

TEXT is very flexible as you can apply many different formats. When you use the & symbol to 

join text together you have to use the TEXT format dates and numbers. 

Excel stores dates as numbers. When you join a date with no formatting the number is 

displayed. See cell C4 in the Joining sheet. 

 

=TEXT(Value,Format) 

Value – is usually a cell reference 

Format - must be enclosed in quotation 

marks. The Formats are the same as those 

found in the Custom section of the Number 

tab in the Format Cells dialog. 

 

Example 
Cell A1 contains 1/1/2011  

=TEXT(A1,”dd mmm yy”)  will display 01 Jan 

11 

 

PRACTICE: 

1. In the Joining sheet column C cell C2 has a simple formula that has been copied 

down. 

 

=A2&" "&B2 

 

2. Note the issues in cells C3 and C4 where you need to apply formats the value and 

the date. 

 

3. The TEXT function is used in the cases. The formulas are as follows D3 is 

 

=A3&" "&TEXT(B3,"$#,###") 

 

Cell D4 has  

 

=A4&" "&TEXT(B4,"mmmm yy")   

Tweaking Text 

Case 
 

If you need to modify the case of the text eg convert to uppercase, lowercase or proper case 

you can use the following three functions UPPER, LOWER and PROPER. All functions take a 

cell reference and convert them to their specific case.  

The UPPER sheet has examples.  

You can use these functions to enclose other formulas to convert the case. 

 

SUBSTITUTE Function 

 

You can use the SUBSTITUTE function as a formula-based alternative to using the Find and 

Replace option. When converting codes between systems sometimes you need to remove 

characters or change characters to enable matching. 

=SUBSTITUTE(Text,Old_Text,New_Text,Instance) 

Text - the text to change, it is usually a cell reference. 

Old_Text - the text to find, enclose in quotation marks if keyed in. 

New_Text - the replacement text, enclose in quotation marks if keyed in. 

Instance - (Optional) - entering 2 would change only the second instance. When omitted, all 

instances are changed.  

 

Examples 

=SUBSTITUTE(A2,"-","") 

Will remove the minus sign from the code in cell A2 

=SUBSTITUTE(A2,"-","/") 

Will replace the minus sign with the forward slash.    

Useful Text Solutions 

Extracting the Sheet Name 
A common request is to extract the current sheet name. Sheet names have to be unique. 

You can name a sheet an account number or a department name. Then you can use a 

formula to automatically change the report based on the sheet name.  

Note: the example below uses a stand-alone file, not the companion Excel file for this 

session. We will build the formula in the Test sheet of the companion file. 

There are two ways to extract the formula. 

Single Formula 
The single formula is quite long, but is self-contained and can be used anywhere. It uses the 

CELL function to extract the sheet name. 

 

CELL Function 

The CELL function returns information about a cell. Note: the CELL function will return a 

blank cell if the file has not been saved. 

=CELL(Information_Type,Reference) 

Information_Type - see image on right for the types of 

information that can be displayed. The “filename” is the most 

common type used. 

Reference - (Optional) - I recommend always using a cell 

reference as I have seen some unusual results when this is 

omitted. 

Example 

=CELL("filename",A1) 

This formula is in cell A1 

but it doesn’t cause a 

circular reference. 

The "filename" option 

displays the full path, full 

filename and the sheet 

name. 

Sheet Name Formula 

 

The formula in cell A1 is 

=RIGHT(CELL("filename",A1),LEN(CELL("filename",A1))-SEARCH("]",CELL("filename",A1))) 

Note: the CELL function is repeated three times. An alternative, shorter formula 

involves having the CELL function in cell A1 and then refer to A1 three times, rather 

than the CELL function.  

If cell A1 contains the CELL function then the much shorter formula is 

=RIGHT(A1,LEN(A1)-SEARCH("]",A1)) 

     

This is not a self-contained formula as it relies on cell A1 having the CELL function in 

it. To use it in other cells you may way want to fix the A1 reference as follows. 

=RIGHT($A$1,LEN($A$1)-SEARCH("]",$A$1)) 

Extracting the File Name  

 

To extract just the file name you could use the formula below. 

=MID(A1,SEARCH("[",A1)+1,SEARCH("]",A1)-SEARCH("[",A1)-1) 

 

Again this is not a self-contained formula as it relies on cell A1 having the CELL 

function. 

 

 

 

 

 


